Lec22C Handouts

April 1, 2021 12:54 PM

Barry Michael

Theorem 11.7 (Trager – Rothstein ~1976)

Let $C, D \in K[x]$, $0 \le \deg C < \deg D$, $\gcd(C,D)=1$, $\operatorname{lc} D = 1$. $\gcd(D,D')=1.$

Let $R(z) = \boxed{\operatorname{res}(C - zD', D, x)} \in K[z]$

Then $\int \frac{C}{D}\,dx = \alpha_1 \ln(v_1) + \cdots + \alpha_k \ln(v_k)$ where

compute and factor into linear factors

(i) α_i are the distinct roots of $R(z)$ and

(ii) $\boxed{v_i = \gcd(C - \alpha_i D', D)}$ (monic gcd)

Note if $L = K(\alpha_i)$ then $v_i \in L[x]$.

Theorem 11.8 (restated)

Suppose F is a field and $\int \frac{C}{D}\,dx = \sum \beta_i \ln(w_i)$ where $\beta_i \in F$ and $w_i \in F[x]$. Then $F \supset L$.

Integration of the logarithmic part using Trager-Rothstein .

```
> c := x^4-3*x^2+6:  d := x^6-5*x^4+5*x^2+4:
> Int(c/d,x);
```

$$\int \frac{x^4 - 3x^2 + 6}{x^6 - 5x^4 + 5x^2 + 4}\, dx = (\ \_\_\ )(\ \_\_\ x\ \_\ )(\ \_\ )(\ \_\ )(\ \_\ ),$$

```
> gcd(d,diff(d,x)); # check that d is square-free
```

$$1$$

```
> Rz := resultant( c-z*diff(d,x), d, x );
```

$$Rz := 2930944\, z^6 + 2198208\, z^4 + 549552\, z^2 + 45796$$

```
> Rz := factor( Rz );
```

$$Rz := 45796 \left(4 z^2 + 1\right)^3$$

```
> alpha := {solve( 4*z^2+1, z )};
```

$$\alpha := \left\{ \frac{1}{2}\,\mathrm{I}, -\frac{1}{2}\,\mathrm{I} \right\}$$ $\alpha_1 = \frac{i}{2}$ $\alpha_2 = -\frac{i}{2}$

```
> v[1] := gcd(c-alpha[1]*diff(d,x),d);
```

$$v_1 := x^3 + x^2\,\mathrm{I} - 3x - 2\,\mathrm{I}$$

```
> v[2] := gcd(c-alpha[2]*diff(d,x),d);
```

$$v_2 := x^3 - \mathrm{I}\,x^2 - 3x + 2\,\mathrm{I}$$

```
> TR := alpha[1]*log(v[1]) + alpha[2]*log(v[2]);
```

$$TR := \frac{1}{2}\,\mathrm{I}\ln\left(x^3 + x^2\,\mathrm{I} - 3x - 2\,\mathrm{I}\right) - \frac{1}{2}\,\mathrm{I}\ln\left(x^3 - \mathrm{I}\,x^2 - 3x + 2\,\mathrm{I}\right)$$

```
> simplify( diff(TR,x)-c/d );
```

$$0$$